Chapter 5

FINANCIAL CALCULATIONS

In This Chapter

- EasyRefresher™: Applying Time Value of Money Concepts
- Using the Standard Financial Functions
- Using the Add-In Financial Functions

Although most Excel users—even most advanced business users—will have scant occasion to use all of the financial functions that Excel provides, almost everybody will use at least a handful of these time-saving functions. Calculating a mortgage or car loan payment requires use of the PMT function. Estimating the future value of an investment requires use of the FV function. Computing the interest rate implicit in a set of loan or investment terms requires the RATE, IRR, or MIRR functions. This chapter explains how you can use any of the financial functions that come with Excel—both those that are always available and those you can use after installing the Analysis ToolPak add-in.

NOTE *The Analysis ToolPak supplies a set of 42 additional financial functions which you use in the same way as the standard financial functions. In general, these add-in financial functions are most useful for security analysis.*

EasyRefresher™: Applying Time Value of Money Concepts

The phrase "time value of money" must surely be one of the most used terms that people don't really understand. Almost invariably, people who don't know a discount rate from Adam use the term to explain or question any financial complexity. The irony—and most business school students and graduates know this—is that time value of money isn't all that complicated. Or at least it isn't at the conceptual level.

Analyzing Borrowing

The time value of money concept, which applies to loans, means that you need to include interest costs in any analysis of loans. In other words, to compare "loan A" to "loan B," you need to account for the interest costs of each. Note that this isn't the same thing as saying you need to compare the interest rates, however. The interest rate of a loan is important. It's the first of the three variables used to calculate the interest charges of a loan. But you need more than just the interest rate to know what, for example, "loan A" costs. You also need to know the loan balance against which the interest rate is applied. (This is the second variable.) And you need to know for how many periods—years, months, or whatever—this calculation is made. (This is the third variable.)

Interestingly, truth-in-lending laws make it easy for consumers to make time value of money comparisons of different borrowing options. By comparing the annual percentage rate, or APR, of one loan with another, one can generally tell which borrowing option is cheapest. The APR wraps all of the costs of borrowing—all the time value of money—into a single, interest-rate-like number. By choosing the lowest APR, a consumer generally gets the cheapest loan. Unfortunately, it isn't as easy for business borrowers to get APR information. (Truth-in-lending laws, for example, apply to consumers but not to business borrowers.) Nevertheless, it's still generally most useful to make time value of money comparisons of different borrowing options by applying the APR concept.

NOTE *In the discussion of the RATE function, I'll describe how to calculate the APR of any loan.*

Analyzing Investments

The time value of money concept also applies to investments, with a slight twist: When applied to investments, you need to factor in the interest or investment earnings generated by an investment. In other words, to compare "investment A" to "investment B," you need to account for the interest or investment returns of each. Note again that isn't the same thing as saying you need to compare the interest rates or investment rates of returns, however. The interest rate or rate of return on an investment is important. As with borrowing comparisons, it's only the first of the three basic variables used to calculate the investment profits from an investment. You also need to know the investment balance against which the interest rate or rate of return is applied. (This is the second variable.) And you need to know for how many periods—years, months, or whatever—this calculation is made. (This is the third variable.)

As a general rule, when people perform time value of money comparisons on investments, they compare either the present values of the two investments or the rates of return of the two investments. The rate comparison method is easier to understand because it works very much like the APRs that lenders provide consumers. To compare "investment A" with "investment B" on the basis of interest rates or investment rates of return, you compare two percentages; whichever is larger is better, so the logic goes.

Comparing investments on the basis of their rates of return, however, creates a handful of problems, as you may already know. First, simple rate comparisons ignore the fact that the investment balance is important. For example, earning a 25% return on a billion dollars is far more profitable than earning a 100% return on a million dollars. Second, simple rate comparisons ignore the rate at which intermediate cash flows are reinvested. If you have 1 million dollars to invest for 10 years and are choosing between one investment which pays 25% for one year and another investment which pays 15% for 10 years, for example, you can't know which is better unless you know the rate at which your money can be reinvested in year two. Third, return-based investment measures sometimes suffer from a sort of mathematical phenomenon in which the return formula can't be solved with a single, unique interest rate or investment rate of return value.

NOTE *It turns out, as mentioned elsewhere in this book, that an investment with "N" cash flows ("N" is number of cash flows) is actually an Nth root polynomial equation with up to "N" real and imaginary solutions to the investment's internal rate of return equation.*

Because of the aforementioned problems with applying the time value of money to investment calculations, business analysts commonly compare investments based on their present values. Two investments' cash flows can be evaluated on an "apples-to-apples" basis by comparing their present values: whichever investment provides the greater present value is better.

You can also compare the present value of an investment's cash flows to its initial cash cost, making what's known as a net present value calculation. A net present value is actually a simple cost benefit comparison. You compare the cost of an investment, meaning its cash price, with its benefits, calculated as the present value of its future cash flows. If the net present value is positive, it means the benefits exceed the cost. If the net present value is negative, it means the cost exceeds the benefit.

A challenging feature of present value and net present value calculations concerns the choice of a discount rate, or interest rate, used to convert future cash flows to their current-day, present value. For example, people argue in favor of using the cost of the capital used to fund an investment. If an investment is made using borrowed money that costs 8%, for example, someone taking this approach might use 8% as the discount rate. Another commonly argued approach is to use the rate of return offered by similarly risky investments. If you can make an investment that forces you to bear the same risk and pays a 12% return, some people argue that you should use a 12% discount. In practice, it's worth mentioning, that one often sees discount rates set almost as a matter of management policy or as an arbitrary decision made by a key participant in the financial analysis process. In a case like this, top management might say, perhaps explicitly, that investments must produce at least a 15% return and that would mean that present value calculations should be made with a 15% discount rate.

Dealing with Inflation

One important issue you want to consider in making time value of money calculations is the effect of inflation. Over time, of course, inflation erodes the value of the currency units used to make time value of money calculations. This erosion makes it difficult to compare currency values of different time periods. For example, 1 million dollars today isn't the same thing as 1 million dollars 20 years from now.

You can, fortunately, rather easily estimate the effect of inflation in your time value of money calculations. To do so, you simply need to use the real rate of return rather than the nominal rate of return in your calculations. You calculate the real rate of return by subtracting the inflation rate from the nominal rate of return.

As an example of how all this works, suppose that you want to estimate the future value of a long-term investment in a stock index fund. Over long periods of time, the stock market returns about 10% and inflation runs about 3.5%, so the real return equals 6.5%. If you used 6.5% in your time value of money calculations—rather than the nominal rate of return of 10%—the future value amounts you'd calculate wouldn't include inflation. In effect, by subtracting the inflation rate from the nominal return, you subtract the effects of the inflation from the compounded, future value amounts you calculate.

Using the Standard Financial Functions

Excel provides 16 standard financial functions for making depreciation, loan payment, present value, future value, and rate of return calculations. To see which financial functions Excel provides or to see which arguments a function requires, choose the Insert menu's Function command and then select Financial from the Function Category list box (see Figure 5-1). The first Paste Function dialog box shows the categories of functions that Excel provides—such as which financial functions are available.



Figure 5-1 The first Paste Function dialog box.

After you select a function and click OK, the second Paste Function dialog box shows which arguments are required for the function to make its calculations (see Figure 5-2).



Figure 5-2 The second Paste Function dialog box.

NOTE *Chapter 2 describes how to work with functions and provide their arguments. If you're not already familiar with how a function makes its calculations, you might want to refer to that chapter.*

Using the Depreciation Functions

Excel supplies five depreciation functions as part of its standard financial function set: DB, DDB, SLN, SYD, and VDB. Each of these functions apportions the cost of a long-lived asset over its estimated economic life.

NOTE *In the examples that follow, the function results are rounded to two decimal places. For example, even if a function returns the value 12804.6875, in the text the function result is reported as 12804.69.*

TIP *Excel also supplies two add-in financial functions for calculating depreciation according to French accounting conventions, AMORDEGRC and AMORTLINC. These two functions are described in the later section "Using the Add-In Financial Functions."*

DB

The DB function calculates fixed declining balance depreciation for an asset given the cost, its salvage value, estimated economic life, the accounting period for which depreciation is being calculated, and, optionally, the number of month in first year. (If you don't include the optional month argument, Excel sets this value to 12.) The DB function uses the following syntax:

```
DB(cost,salvage,life,period,month)
```

Suppose, for example, that you must calculate the fixed declining balance depreciation for equipment that costs $50,000, lasts five years, will have a salvage value of $10,000 at the end of the fifth year, and that was placed into service in the third month of the first year. To calculate the depreciation for the first year, you use the following formula:

```
=DB(50000,10000,5,1,3)
```

The function returns the value 3437.5. To calculate the depreciation for the second year, you use the formula

```
=DB(50000,10000,5,2,3)
```

The function returns the value 12804.69

The distinguishing feature of fixed-declining balance depreciation is that it calculates depreciation at a fixed rate based on the estimated cost, salvage value, and economic life of the asset. Excel calculates this rate using the following formula:

```
Fixed rate=1-((salvage/cost)^(1/life))
```

and then rounds this value to the nearest three decimal places. To calculate the depreciation for a period, Excel multiplies the rate by the sum of the original cost less the accumulated depreciation to date.

NOTE *The accumulated depreciation equals the original cost minus the previous periods' depreciation.*

Excel uses variations of the standard fixed-declining balance formula for the first and last periods. For the first period, Excel calculates the depreciation by using the following formula:

```
First-period depreciation=cost * rate * month / 12
```

For the last period, Excel calculates the depreciation using the following formula (which essentially just depreciates the asset down to its salvage value):

```
Last-period depreciation=((cost-accumulated depreciation)*rate*(12-
month))/12
```

DDB

The DDB function calculates double-declining balance depreciation for an asset given the cost, its salvage value, estimated economic life, the accounting period for which depreciation is being calculated, and, optionally, the factor at which the balance declines. (If you don't include the optional factor argument, Excel sets this value to 2 indicating "double" declining balance.) The DDB function uses the following syntax:

```
DDB(cost,salvage,life,period,factor)
```

Suppose, for example, that you must calculate the double-declining balance depreciation for equipment that costs $50,000, lasts five years, and will have a salvage value of $10,000 at the end of the fifth year. To calculate the depreciation for the first year, you use the following formula:

```
=DDB(50000,10000,5,1)
```

The function returns the value 20000.00. To calculate the depreciation for the second year, you use the formula

```
=DDB(50000,10000,5,2)
```

The function returns the value 12000.00.

NOTE *A common convention when using double-declining balance depreciation is to switch to straight-line depreciation at the point in time when straight depreciation exceeds declining balance depreciation. The DDB function doesn't make this switch, but the VDB function does. Use it, therefore, if you want to use this convention.*

SLN

The SLN function calculates straight-line balance depreciation for an asset given the cost, its salvage value, and its estimated economic life. The DDB function uses the following syntax:

```
SLN(cost,salvage,life)
```

Suppose, for example, that you must calculate the straight-line depreciation for equipment that costs $50,000, lasts five years, will have a salvage value of $10,000 at the end of the fifth year. To calculate the depreciation for the first year, you use the following formula:

```
=SLN(50000,10000,5)
```

The function returns the value 8000.00. To calculate the depreciation for the second year, you use the same formula because straight-line depreciation is the same for each year period.

SYD

The SYD function calculates sum-of-the-years-digits depreciation for an asset given the cost, its salvage value, estimated economic life, and the accounting period for which depreciation is being calculated. The SYD function uses the following syntax:

```
SYD(cost,salvage,life,period)
```

Suppose, for example, that you must calculate the sum-of-the-years-digits depreciation for equipment that costs $50,000, lasts five years, and will have a salvage value of $10,000 at the end of the fifth year. To calculate the depreciation for the first year, use the following formula:

```
=SYD(50000,10000,5,1)
```

The function returns the value 13333.33. To calculate the depreciation for the second year, you use the formula

```
=SYD(50000,10000,5,2)
```

The function returns the value 10666.67.

VDB

The VDB function calculates declining balance depreciation for an asset given the cost, its salvage value, estimated economic life, the starting accounting period and the ending accounting period for which depreciation is being calculated, the factor at which the balance declines, and, optionally, a switch-to-straight-line switch which is set to either TRUE or FALSE. If you set this switch to TRUE, Excel doesn't switch to straight-line at the point when straight-line depreciation exceeds declining balance depreciation. If you set this value to FALSE, Excel does switch to straight-line. If you don't set the optional switch-to-straight-line switch to TRUE, Excel sets this value to FALSE.

The VDB function uses the following syntax:

```
VDB(cost,salvage,life,start period,end period,factor,switch)
```

Suppose, for example, that you must calculate 150% declining balance depreciation for equipment that costs $50,000, lasts five years, and will have a salvage value of $10,000 at the end of the fifth year. To calculate the depreciation for the first year, you use the following formula:

```
=VDB(50000,10000,5,0,1,150%)
```

The function returns the value 15000.00. Notice that to calculate depreciation for the first year, you set the start period to 0 and the end period to 1. To calculate the depreciation for the second year, you use the formula

```
=VDB(50000,10000,5,1,2,150%)
```

The function returns the value 10500.00. Notice that to calculate the depreciation for the second year, you set the start period to 1 and the end period to 2.

In both of the two preceding examples, Excel will automatically switch to straight-line depreciation at the point when straight-line depreciation for a period exceeds declining balance depreciation. To instruct Excel not to make this switch, you would use the following formula to calculate depreciation for the first year:

```
=VDB(50000,10000,5,0,1,150%,TRUE)
```

The word TRUE, which Excel interprets as 1, tells Excel not to switch to straight-line. To instruct Excel not to make this switch in the second year, you would use the following formula to calculate depreciation:

```
=VDB(50000,10000,5,1,2,150%,TRUE)
```

Using the Payment Functions

Excel provides five standard payment functions: IPMT, ISPMT, PMT, PPMT, and NPER. Typically, you use these functions to calculate loan payment information. You can also use them for investment annuity calculations. The paragraphs that follow describe each of these payment functions and give examples of each. As you work with each, however, keep two factors in mind:

- Be sure that you stay consistent in your period assumptions between the payment and the term and rate. In other words, if you work with monthly payments, your term and interest rate must also be expressed as monthly amounts. And if you work with annual payments, your term and interest rate must also be expressed as annual amounts.
- Note that you must use the sign of a value to indicate whether it is a cash inflow or cash outflow. An initial loan balance—assuming you're the borrower—should be shown as a positive value because it represents a cash inflow. And loan payments as well as any balloon payments—again, assuming you're the borrower—should be shown as negative values because they represent cash outflows. Note, too, that Excel uses signs of values in the same ways. It shows cash inflows as positive values and cash outflows as negative values.

You must keep both of these factors in mind as you work with the financial functions and especially as you work with the payment functions.

NOTE *In keeping with the naming conventions employed by Excel, this book uses* rate *to refer to the periodic interest rate,* period *to refer to the payment period,* pmt *to refer to the regular payment or annuity amount,* nper *to refer to the term,* pv *to refer to the present value, and* fv *to refer to the future value.*

IPMT

The IPMT function calculates the interest portion of a payment given its interest rate, the period, the term (or number of payments), present value (or loan balance), future value (or balloon payment), and, optionally, the type-of-annuity switch. If you set the type-of-annuity switch to 1, Excel assumes payments occur at the beginning of the period, following the annuity due convention. If you set the annuity switch to 0 or you omit the argument, Excel assumes payments occur at the end of the period following the ordinary annuity convention.

The function uses the following syntax:

```
IPMT(rate,period,nper,pv,fv,type)
```

For example, to calculate the period interest rate for the 54th payment on a 30-year, $150,000 mortgage charging 8% annual interest, you use the following formula:

```
=IPMT(.08/12,54,30*12,150000,0,0)
```

The function returns the value -957.51. Notice that to convert the 8% annual interest to a period interest, the formula divides the annual interest rate by 12. Notice, too, that to convert the 30-year term to a term in months, the formula multiplies 30 by 12. The function returns the interest payment amount as a negative value because it reflects a cash outflow you pay.

NOTE *If you set the* pv *argument to –150000, you indicate that you're loaning money. In this case, the function returns 957.51, a positive value, showing that the interest payment amount is a positive cash inflow.*

ISPMT

Provided for compatibility with Lotus 1-2-3, the ISPMT function calculates the straight-line interest portion of a payment given its interest rate, the period, the term (or number of payments), and present value (or loan balance).

```
ISPMT(rate,period,nper,pv)
```

For example, to calculate the period interest rate for the 54th payment on a 30-year, $150,000 mortgage charging 8% annual interest, you use the following formula:

```
=ISPMT(.08/12,54,30*12,150000)
```

The function returns the value –850.00. Notice that to convert the 8% annual interest to a period interest, the formula divides the annual interest rate by 12. Notice, too, that to convert the 30-year term to a term in months, the formula multiplies 30 by 12. The function returns the interest payment amount as a negative value because it reflects a cash outflow you pay.

PMT

The PMT function calculates a payment given its interest rate, the term (or number of payments), present value (or loan balance), future value (or balloon payment), and, optionally, the type-of-annuity switch. If you set the type-of-annuity switch to 1, Excel assumes payments occur at the beginning of the period, following the annuity due convention. If you set the annuity switch to 0 or you omit the argument, Excel assumes payments occur at the end of the period following the ordinary annuity convention.

The function uses the following syntax:

```
PMT(rate,nper,pv,fv,type)
```

For example, to calculate the payment on a 30-year, $150,000 mortgage charging 8% annual interest, you use the following formula:

```
=PMT(.08/12,30*12,150000,0,0)
```

The function returns the value –1100.65. Notice that to convert the 8% annual interest to a period interest, the formula divides the annual interest rate by 12. Notice, too, that to convert the 30-year term to a term in months, the formula multiplies 30 by 12. The function returns the interest payment amount as a negative value because it reflects a cash outflow you pay.

NOTE *If you set the* pv *argument to –150000, you indicate that you're actually loaning money. And in this case, the function returns 1100.65, a positive value, showing that the payment amount is a positive cash inflow.*

For example, to calculate the payment on a 30-year, $150,000 mortgage charging 8% annual interest that has a $25,000 balloon payment, you use the following formula:

```
=PMT(.08/12,30*12,150000,-25000,0)
```

The function returns the value -1083.87. Notice that the balloon payment argument appears as a negative value because it represents a cash outflow.

NOTE *If you set the* fv *argument to 25000, you indicate that you're actually receiving a final payment from the lender. In this case, the function returns –1117.42, which is larger than required to pay off the loan. The extra payment, in effect, gets invested at the loan interest rate and future values to the $25,000.*

NOTE *You can easily calculate the total payments made on a loan. To do this, multiply the payment amount, as calculated by the PMT function, by the number of payments. Note that once you have this result, you can also easily calculate the total interest paid on a loan, too. To do this, subtract the loan balance from the total payments.*

While you'll most typically use the PMT function to calculate loan payments, you can also use it to calculate the payment required to accumulate some future value. Suppose, for example, that you want to calculate how large a contribution an employee would need to make into a 401(k) account in order to amass a $1,000,000 portfolio over 35 years. If you assume the employee will earn 9% annually, you use the following formula to make this estimate:

```
=PMT(.09,35,0,1000000,0)
```

The function returns the value –4635.84. Notice that the type switch is 0, which means that function returns the amount that must be paid at the end of the year.

If you instead want to calculate the amount that would need to be paid at the beginning of each month, you would use the following formula to make this estimate:

```
=PMT(.09/12,35*12,0,1000000,1)
```

This formula returns the value –337.40. This value is slightly less than one-twelfth of the annual, ordinary annuity value because by making payments throughout the year at the start of each month, the employee earns additional interest.

If you wanted to make the same calculation but also recognize the added fact that the employee already has $10,000 in her 401(k) account, you would use the formula:

```
=PMT(.09/12,35*12,-10000,1000000,1)
```

This formula returns the value -259.58.

PPMT

The PPMT function calculates the principal portion of a payment given its interest rate, the period, the term (or number of payments), present value (or loan balance), future value (or balloon payment), and, optionally, the type-of-annuity switch. If you set the type-of-annuity switch to 1, Excel assumes payments occur at the beginning of the period, following the annuity due convention. If you set the annuity switch to 0 or you omit the argument, Excel assumes payments occur at the end of the period following the ordinary annuity convention.

The function uses the following syntax:

```
PPMT(rate,period,nper,pv,fv,type)
```

For example, to calculate the period principal payment for the 54th payment on a 30-year, $150,000 mortgage charging 8% annual interest, you use the following formula:

```
=PPMT(.08/12,54,30*12,150000,0,0)
```

The function returns the value –143.13. Notice that to convert the 8% annual interest to a period interest, the formula divides the annual interest rate by 12. Notice, too, that to convert the 30-year term to a term in months, the formula multiplies 30 by 12. The function returns the principal payment amount as a negative value because it reflects a cash outflow you pay.

NOTE *If you set the* pv *argument to –150000, you indicate that you're actually loaning money. And in this case, the function returns 143.13, a positive value, showing that the principal payment amount is a positive cash inflow.*

NPER

The NPER function calculates the term, or number of regular payments, on a loan or for an investment annuity given its interest rate, the payments, present value (or loan balance), future value (or balloon payment), and, optionally, the type-of-annuity switch. If you set the type-of-annuity switch to 1, Excel assumes payments occur at the beginning of the period, following the annuity due convention. If you set the annuity switch to 0 or you omit the argument, Excel assumes payments occur at the end of the period following the ordinary annuity convention.

The function uses the following syntax:

```
NPER(rate,pmt,pv,fv,type)
```

For example, to calculate the number of $1,000 monthly payments required to pay off a 9% mortgage that still has a $100,000 mortgage balance, you use the following formula:

```
=NPER(.09/12,-1000,100000,0,0)
```

The function returns the value 185.53, representing roughly 185 payments and then another roughly half payment. Notice that to convert the 9% annual interest to a period interest, the formula divides the annual interest rate by 12. Notice, too, that the payment amount, as a cash outflow, shows as a negative value while the loan balance, as an implicit cash inflow, shows as a positive value.

NOTE *The NPER function rarely returns an integer, or whole-number result. As in the preceding example, it commonly returns a fractional value, indicating that after the last regular payment, an additional fractional payment will also need to be made.*

You can also use the NPER function to calculate investment terms. In this case, you calculate the number of payments that need to be made in order to reach some future value. Suppose, for example, that you want to calculate how many years a customer needs to contribute $2,000 to an Individual Retirement Account in order to amass a $1,000,000 portfolio. If you assume the customer will earn 9% annually and will make payments at the beginning of the year, you use the following formula to make this estimate:

```
=NPER(.09,-2000,0,1000000,1)
```

The function returns the value 43.45, indicating the $2,000 payments will need to be made for slightly more than 43 years. Notice that the type switch is 1, which means that the function returns the amount that must be paid at the beginning of the year. If you instead want to calculate the amount that would need to be paid at the beginning of each year, you would use the following formula to make this estimate:

```
=NPER(.09,-2000,0,1000000,0)
```

This formula returns the value 44.43. This value is slightly more than the annuity due value because by making payments at year-end, the customer loses interest.

If you wanted to make the same calculation but recognize the added fact that the customer already has $5,000 in his IRA account, you would use the formula:

```
=NPER(.09,-2000,-5000,1000000,0)
```

This formula returns the value 42.07.

Using the Present Value, Future Value, and Interest Rate Functions

Excel rounds out its standard financial function set with six additional functions for calculating present values, future values, and interest rates, or rates of return, including FV, IRR, MIRR, NPV, PV, and RATE.

The paragraphs that follow describe each of these functions and give examples of each. As you work with each function, let me reiterate that you need to keep two factors in mind:

- Be sure that you stay consistent in your period assumptions between the payment and the term and rate. In other words, if you work with monthly payments, your term and interest rate must also be expressed as monthly amounts. And if you work with annual payments, your term and interest rate must also be expressed as annual amounts.
- Note that you must use the sign of a value to indicate whether it is a cash inflow or cash outflow. A present value—assuming you're the investor—should be shown as a negative value because it represents a cash outflow. And annuity amounts as well as any balloon payments—again, assuming you're the investor—should be shown as positive if they represent cash inflows and as negative values if they represent cash outflows. Note, too, that Excel uses signs of values in the same ways. It shows cash inflows as positive values and cash outflows as negative values.

NOTE *In keeping with the naming conventions employed by Excel, this book uses* rate *to refer to the periodic interest rate,* period *to refer to the payment period,* pmt *to refer to the regular payment or annuity amount,* nper *to refer to the term,* pv *to refer to the present value, and* fv *to refer to the future value.*

FV

The FV function calculates the future value of a loan or investment given its interest rate, the term (or number of payments), the payment, the present value (or loan balance), and, optionally, the type-of-annuity switch. If you set the type-of-annuity switch to 1, Excel assumes payments occur at the beginning of the period, following the annuity due convention. If you set the annuity switch to 0 or you omit the argument, Excel assumes payments occur at the end of the period following the ordinary annuity convention.

The function uses the following syntax:

```
FV(rate,nper,pmt,pv,type)
```

For example, to calculate the future value of a $200-a-month savings program over 25 years assuming that the investor starts with $10,000 and earns 10% annual interest, you use the following formula:

```
=FV(10%/12,25*12,-200,-10000,0)
```

The function returns the value 385936.13. Notice that to convert the 10% annual interest to a monthly interest rate, the formula divides the annual interest rate by 12. Notice, too, that to convert the 25-year term to a term in months, the formula multiplies 25 by 12. The monthly payment and initial present values show as negative amounts because they represent cash outflows. And the function returns the future value amount as a positive value because it reflects a cash inflow the investor ultimately receives.

You can also use the FV function to estimate loan balloon payment amounts. Suppose, for example, that you want to calculate the balloon payment required to pay off a $150,000 mortgage with an 8% annual interest rate after the buyer has been making $1,200-a-month payments for 10 years. You use the following formula to make this estimate:

```
=FV(8%/12,10*12,-1200,150000,0)
```

The function returns the value −113410.79. Notice that the interest rate is divided by 12 and the number of years of payments is multiplied by 12 to adjust these figures to monthly amounts. Also, notice that the monthly payment amount shows as a negative value to show it represents a cash outflow, and the initial loan balance shows as a positive value to show that it represents a cash inflow.

NOTE *Excel also supplies an FVSCHEDULE function, which lets you calculate a future value using varying annual interest rates. The FVSCHEDULE function is described in the later section "Using the Future Value Add-In Functions."*

IRR

The IRR calculates the internal rate of return implicit in a set of cash flows given a values argument (usually a worksheet range holding the cash flow values), and, optionally, a guess at the internal rate of return value. The internal rate of return of a set of cash flows is the discount rate that produces a net present value equal to zero.

NOTE *Excel also provides a net present value function, NPV.*

The function uses the following syntax:

```
IRR(values,guess)
```

For example, if you store the cash flows from an investment into worksheet like the one shown in Figure 5-3, you can use the following formula to calculate the investment's internal rate of return:

```
=IRR(B1:B11)
```

The function returns the value 27.13%.

	A	B	C
1	Initial investment	-25000	
2	Year 1 cash flow	8000	
3	Year 2 cash flow	7000	
4	Year 3 cash flow	8000	
5	Year 4 cash flow	8000	
6	Year 5 cash flow	8000	
7	Year 6 cash flow	6000	
8	Year 7 cash flow	7000	
9	Year 8 cash flow	7000	
10	Year 9 cash flow	7000	
11	Year 10 cash flow	5000	
12			
13	IRR	27.13%	
14			

Figure 5-3 A simple worksheet illustrating how an IRR function works.

You'll want to consider several things if you use the IRR function:

- The values argument needs to contain at least one positive value and at least one negative value. If your investment doesn't meet at least these requirements, it doesn't look enough like an investment to be measured by the IRR function.
- The order of the cash flows in the values argument should reflect their actual order: the first cash flow first, the second cash flow second, and so on.

- The cash flow periods must be consistent. In Figure 5-3, for example, the cash flow periods are all years—and that works. But you couldn't mix and match annual and monthly cash flow in such a values range.

NOTE *If your values range includes a cell that includes a label, a logical value, or an empty cell, that cell is ignored in the IRR calculations.*

Although you aren't required to use a guess argument because Excel assumes an initial guess of 10%, you may need to do so. Excel attempts to solve the IRR function's equation iteratively. If the equation can't be solved after 20 attempts to within .00001, the function returns the #NUM error value. Excel usually succeeds if the internal rate of return value is close to typical returns (say, between –10% and 30%), but it sometimes has trouble with returns that are outside this range. And in this case, if you supply a guess argument that's close to the actual internal rate of return value, you in effect help Excel start its search close to its destination.

One consideration in using the IRR function is that theoretically it doesn't actually have a single correct solution. In practice, a set of cash flows will sometimes have as many valid internal rates of return as there are sign changes.

In the worksheet shown in Figure 5-3, only one cash flow sign change occurs (from negative cash flow at the start of the investment to positive cash flow in year one). In a case like that shown in Figure 5-4, however, the IRR function can actually return two valid internal rate of return values because there are two sign changes—one between the initial investment and the first-year cash flow and another between the first-year cash flow and the second-year cash flow.

	A	B	C	D
1	Initial investment		-50000	
2	Year 1 cash flow		150000	
3	Year 2 cash flow		-100000	
4				
5	IRR if no guess		0%	
6				
7	IRR if guess is 50%		100%	
8				

Figure 5-4 A simple worksheet illustrating how an IRR function can sometimes malfunction.

If you don't supply a guess, Excel calculates the internal rate of return to be equal to zero. If you happen to supply a large guess value—the worksheet in Figure 5-4 uses a guess equal to 50%—Excel calculates the internal rate of return to be equal to 100%. Both calculations are right. What's happening is that Excel finds the calculation result that's closest to your guess or its guess.

NOTE *Excel also supplies an XIRR function, which lets you calculate an internal rate of return for an investment with irregular cash flows but without having to construct a worksheet schedule of the cash flows. The XIRR function is described in the later section "Using the Capital Budgeting Add-In Functions."*

MIRR

The MIRR function calculates the modified internal rate of return implicit in a set of cash flows given the cash flow values (usually a worksheet range holding the cash flow values), the finance rate, and the reinvestment rate. This rate of return calculation differs from the IRR function's calculation (described in the preceding paragraphs) in two ways: it assumes that interim cash inflows are reinvested at some specified rate of return, and it assumes that interim cash outflows are funded by borrowing at some specified interest rate.

NOTE *Unlike the IRR function, you should be able to calculate a single, unique return measure using the MIRR function—as long as the function arguments make sense.*

The function uses the following syntax:

```
MIRR(values,finance rate,reinvest rate)
```

For example, if you store the cash flows from an investment into a worksheet such as the one shown in Figure 5-5, you will borrow money at 10% to pay for the first year's cash outflow, and then you will reinvest the cash flows in years two through ten at 12%. You can use the following formula to calculate the investment's modified rate of return:

```
=MIRR(C1:C11,.1,.12)
```

The function returns the value 12.43%.

	A	B	C	D
1	Initial investment		-100000	
2	Year 1 cash flow		-50000	
3	Year 2 cash flow		30000	
4	Year 3 cash flow		35000	
5	Year 4 cash flow		30000	
6	Year 5 cash flow		25000	
7	Year 6 cash flow		30000	
8	Year 7 cash flow		35000	
9	Year 8 cash flow		35000	
10	Year 9 cash flow		35000	
11	Year 10 cash flow		35000	
12				
13	Modified IRR		12.43%	
14				

Figure 5-5 A simple worksheet illustrating how a MIRR function works.

While it's not clear from the Excel documentation that this is the case, MIRR discounts any interim cash outflows back to an equivalent present value using the finance rate and compounds any interim cash inflows out to an equivalent future value amount using the reinvestment rate. The modified internal rate of return value, then, is the rate that equates the initial present value of the cash outflows with the future value of the cash inflows.

As with the IRR function, you'll want to consider several factors if you use the MIRR function:

- The values argument needs to contain at least one positive value and at least one negative value. If your investment doesn't meet at least these requirements, it doesn't look enough like an investment to be measured by the MIRR function.
- The order of the cash flows in the values argument should reflect their actual order: the first cash flow first, the second cash flow second, and so on.
- The cash flow periods must be consistent. In Figure 5-5, for example, the cash flow periods are all years.

NOTE *If your values range includes a cell that includes a label, a logical value, or an empty cell, that cell is ignored in the IRR calculations.*

NPV

The NPV function calculates the net present value of a set of cash flows given the discount rate and the cash flow values (usually a worksheet range holding the cash flow values). If you are using the NPV function to compare alternative investments, the investment opportunity with the largest NPV is the one that generates the largest profit in absolute, present value terms.

The function uses the following syntax:

```
NPV(rate,values)
```

For example, if you store the cash flows from an investment into a worksheet such as the one shown in Figure 5-6 (with the same cash flows as those shown in Figure 5-5), you can use the following formula to calculate the investment's modified rate of return:

```
=NPV(C13,C1:C11)
```

The function returns the value $5,798.18.

	A	B	C	D
1	Initial investment		-100000	
2	Year 1 cash flow		-50000	
3	Year 2 cash flow		30000	
4	Year 3 cash flow		35000	
5	Year 4 cash flow		30000	
6	Year 5 cash flow		25000	
7	Year 6 cash flow		30000	
8	Year 7 cash flow		35000	
9	Year 8 cash flow		35000	
10	Year 9 cash flow		35000	
11	Year 10 cash flow		35000	
12				
13	Discount Rate		12.00%	
14				
15	Net Present Value		$5,798.18	
16				

Figure 5-6 A simple worksheet illustrating how an NPV function works.

You'll want to consider several factors if you use the NPV function:

- The NPV function assumes that the first cash flow occurs immediately, or at what's sometimes referred to as period 0.
- The rate argument needs to be the discount rate for the time period used to calibrate cash flows. In other words, if you're using monthly cash flows, your discount rate needs to be a monthly rate.
- The values argument can include more than one cell or range. You can, for example, use the NPV function NPV(.1,A1,A2,A3,A4:A8).
- The order of the cash flows in the values argument should reflect their actual order: the first cash flow first, the second cash flow second, and so on.
- The NPV function recognizes empty cells, or cells that contain text labels are zeroes in its calculations. However, if you include an array, any empty cells or cells containing text labels are ignored.

NOTE *The NPV function and the IRR functions are related and, in fact, use the same formula (although in different ways). In effect, what the IRR function does is calculate the discount rate at which the net present value equals zero. This discount rate is the internal rate of return.*

NOTE *Excel also supplies an XNPV function, which lets you calculate a net present value for an investment with irregular cash flows but without having to construct a worksheet schedule of the cash flows. The XNPV function is described in the later section "Using the Capital Budgeting Add-In Functions."*

PV

The PV function calculates the present value of an annuity, or future value, given the periodic rate, number of periods, payment, future value (or balloon payment), and, optionally, the type-of-annuity switch. If you set the type-of-annuity switch to 1, Excel assumes payments occur at the beginning of the period, following the annuity due convention. If you set the annuity switch to 0 or you omit the argument, Excel assumes payments occur at the end of the period following the ordinary annuity convention.

The function uses the following syntax:

```
PV(rate,nper,pmt,fv,type)
```

For example, if you want to estimate the outstanding balance on a mortgage loan that charges 8%, requires two hundred more $1,000-a-month payments, and also requires a $10,000 balloon payment, you can use the following formula:

```
=PV(.08/12,200,-1000,-10000)
```

The function returns the value $112,932.75.

NOTE *You must include either the payment (or pmt) argument or the future value (or fv) argument in order to calculate the present value. The PV function, predictably, needs something—either a payment stream or a future value—to calculate the present value of.*

RATE

The Rate function calculates the interest rate implicit in a set of loan or investment terms given the number of periods, the payment, the present value, the future value, and, optionally, the type-of-annuity switch, and also optionally, an interest-rate rate.

If you set the type-of-annuity switch to 1, Excel assumes payments occur at the beginning of the period, following the annuity due convention. If you set the annuity switch to 0 or you omit the argument, Excel assumes payments occur at the end of the period following the ordinary annuity convention.

The function uses the following syntax:

```
RATE(nper,pmt,pv,fv,type,guess)
```

For example, suppose you want to calculate the implicit interest rate on a car lease that requires five years of $250-a-month payments (occurring as an annuity due) and also a $15,000 balloon payment. To do this, assuming you want to start with a guess of 10%, you can use the following formula:

```
=RATE(5*12,-250,20000,-15000,1)
```

The function returns the value .95%, which is a monthly interest rate of just less than 1%. If you annualize this monthly rate by multiplying it by 12, you get an equivalent annual interest rate of 11.41%.

NOTE *Excel solves the RATE function iteratively starting with the guess argument you provide. (If you don't provide this optional argument, Excel uses 10%.) If Excel can't solve the RATE argument within 20 attempts, it returns the #NUM! error. You can try a different guess argument, which may help because you're telling Excel to begin its search from a different starting point.*

Using the Add-In Financial Functions

In addition to Excel's 16 standard financial functions, Excel users also have access to another 42 add-in financial functions. Most business users of Excel won't need these add-in functions, but if you perform much financial analysis with Excel—especially financial analysis of investments—you'll find frequent occasion to use these tools.

NOTE *To use the add-in financial functions, you may need to install the Analysis ToolPak add-in. To install the Analysis ToolPak, choose the Tools menu's Add-Ins command, select the Analysis ToolPak check box, and click OK. Excel may prompt you to insert the Office 2000 or Excel 2000 CD if it isn't currently in your CD-ROM drive.*

Once you've installed the Analysis ToolPak, you can work with the add-in financial functions. You use these functions in the same way that you use other functions, including the financial functions described in the earlier pages of this chapter. The paragraphs that follow describe each of these add-in functions.

Using the Accrued Interest Add-In Functions

Excel provides two functions that help you with accrued interest calculations for securities that pay interest. ACCRINT calculates accrued interest for a security (such as a bond) that pays periodic interest. ACCRINTM calculates accrued interest for a security (such as a zero-coupon bond) that pays interest upon maturity.

The accrued interest functions use a similar set of arguments, including the issue date, first interest date, settlement date, maturity date, coupon rate, par value, frequency, and basis.

The date arguments are self-explanatory for the most part. The issue date is the date the security is issued. The first interest payment date is the first coupon date. The settlement date is the date you purchased, or settled, the bond. The maturity date is the date the bond matures, or expires. You may enter the date arguments either as text strings enclosed in quotation marks (for example, "7/4/99") or as serial date values (for example, 37000 for April 19, 2001).

The coupon rate and par value arguments let Excel calculate the interest. The coupon rate is the annual interest rate multiplied by the par value to calculate the annual interest. For example, if a bond pays 8% interest annually and the par value is $1,000, Excel would calculate the annual interest by multiplying the 8% by the $1,000 if the coupon is paid annually.

The frequency argument gives the number of coupon payments made each year: you specify 1 to indicate an annual coupon, 2 to indicate a semiannual coupon, and 4 to indicate a quarterly coupon.

The basis argument specifies the number of days in the month and in the year assumed for the date calculations. You specify the basis as 0 for the US (or NASD) version of 30 days in a month and 360 days in a year; as 1 for the actual number of days in the month and year; 2 for the actual number of days in the month but 360 days in a year; 3 for the actual number of days in the month and 365 days in a year; and 4 for the European version of 30 days in a month and 360 days in a year.

Both the ACCRINT and ACCRINTM functions return an error value in the following situations:

- If you enter an invalid date argument, Excel returns #VALUE.
- If the coupon rate or par value argument is less than 0, Excel returns #NUM.
- If the payment frequency is some number other than 1, 2, or 4, Excel returns #NUM.
- If the day-count-basis switch isn't 1, 2, 3, or 4, Excel returns #NUM.
- If issue date follows the settlement date, Excel returns #NUM.

ACCRINT

The ACCRINT function calculates the accrued interest for a security that pays periodic interest given the issue date, first interest payment date, settlement date, coupon rate, par value, payment frequency, and a day-count-basis switch. It uses the following syntax:

```
ACCRINT(issue,first  interest,settlement,rate,par,frequency,basis)
```

For example, if you want to calculate the accrued interest on a bond that was issued on February 8, 1999, first paid interest on April 8, 1999, was purchased on May 23, 2000, pays an 8% coupon, shows a $1,000 par value, pays interest four times a year, and uses the US, or NASD, day-count-basis assumption, you use the following formula:

```
=ACCRINT("2/8/99","4/8/99","5/23/00",0.08,1000,4,0)
```

The function returns the value 103.33.

ACCRINTM

The ACCRINTM function calculates the accrued interest for a security that pays interest at maturity given the issue date, the maturity date, coupon rate, par value, and a day-count-basis switch. It uses the following syntax:

```
ACCRINTM(issue,maturity,rate,par,basis)
```

For example, if you want to calculate the accrued interest on a bond that was issued on February 8, 1991, matures on May 23, 2010, accrues an 8% coupon, shows a $1,000 par value, accrues interest two times a year, and uses the US, or NASD, day-count-basis assumption, you use the following formula:

```
=ACCRINTM("2/8/91","5/23/10",0.08,1000,2)
```

The function returns the value 1565.33.

Using the Bond Duration Add-In Functions

Excel provides two functions that let you make bond duration calculations: DURATION and MDURATION. Duration, a weighted average measure of the present value of a bond's cash flows, quantifies how a change in the bond yield affects the bond price.

Both duration functions use the same set of six arguments: the settlement date, the maturity date, the coupon rate, the yield, the coupon frequency, and the day count basis.

The settlement date specifies the date the bond is settled, or purchased. The maturity date specifies the date the bond matures, or expires. As with the other add-in financial functions, you may enter the date arguments either as text strings enclosed in quotation marks or as serial date values.

The coupon rate argument is the bond's interest rate and is used to calculate coupon payments. The yield argument is the bond's annual yield.

NOTE *Both duration functions assume that the bond's face, or par, value equals $100.*

The frequency argument gives the number of coupon payments made each year: you specify 1 to indicate an annual coupon, 2 to indicate a semiannual coupon, and 4 to indicate a quarterly coupon.

The basis argument specifies the number of days in the month and year assumed for the date calculations. You specify the basis as 0 for the US (or NASD) version of 30 days in a month and 360 days in a year; as 1 for the actual number of days in the month and actual number of days; 2 for the actual number of days in the month but 360 days in a year; 3 for the actual number of days in the month and 365 days in a year; and 4 for the European version of 30 days in a month and 360 days in a year.

NOTE *Excel uses only the integer portion of the arguments you supply to the add-in price and yield date functions. If you enter an argument with decimal values, Excel truncates the argument to just its integer component.*

The duration functions return an error value in several predictable cases:

- If you use an invalid date, Excel returns #VALUE. Note that this means your date arguments must make sense collectively, too. For example, your maturity date must follow the settlement date.
- If you use a frequency argument other than 1, 2, or 4, Excel returns #NUM.
- If you use a day-count-basis switch other than 0, 1, 2, 3, or 4, Excel returns #NUM.
- If the settlement day follows the maturity date, Excel returns #NUM.
- If the rate or yield is less than zero, Excel returns #NUM.

DURATION

The DURATION function calculates a Macauley duration given the settlement date, maturity date, coupon rate, yield, frequency, and basis. It uses the following syntax:

```
DURATION(settlement,maturity,coupon,yield,frequency,basis)
```

For example, suppose you want to calculate the duration of a bond you purchased on April 23, 2000, and that will mature on November 30, 2020. Further suppose that the coupon rate is 8%, which is paid in four quarterly payments, but that the bond yield is 7%. If you want to use the US (NASD) day count basis of 30 days in a month and 360 days in a year, you would use the following formula to calculate this bond's yield:

```
=DURATION("4/23/2000","11/30/2020",.08,.07,4,0)
```

The formula returns the value 10.6496.

MDURATION

The MDURATION function calculates a modified duration given the settlement date, maturity date, coupon rate, yield, frequency, and basis. It uses the following syntax:

```
MDURATION(settlement,maturity,coupon,yield,frequency,basis)
```

For example, suppose you want to calculate the duration of a bond you purchased on April 23, 2000, and that will mature on November 30, 2020. Further suppose that the coupon rate is 8%, which is paid in four quarterly payments, but that the bond yield is 7%. If you want to use the US (NASD) day count basis of 30 days in a month and 360 days in a year, you would use the following formula to calculate this bond's yield:

```
=MDURATION("4/23/2000","11/30/2020",.08,.07,4,0)
```

The formula returns the value 10.4664.

Using the Capital Budgeting Add-In Functions

Excel provides standard functions, IRR and NPV, for calculating the internal rate of return and net present value of a set of cash flows. While most often you'll want to use these two functions, they may sometimes present a practical problem: Both the IRR and NPV functions assume you've first constructed a worksheet that arranges the cash flows into equal periods. In other words, to use the IRR or NPV function, you must first construct a worksheet that shows the investment's monthly cash flows, or its annual cash flows, or the cash flows from some other consistent time period.

Unlike the IRR and the NPV functions, the XIRR and XNPV functions don't require you to first construct a worksheet schedule that arranges the investment cash flows into equal periods. With the XIRR and XNPV functions, you supply the date values that correspond to the cash flow values to the function as arguments.

NOTE *You might want to review this chapter's earlier discussion of the IRR and NPV functions if you have questions about how these two capital budgeting tools work.*

The somewhat unique feature of both the XIRR and the XNPV function is that if you supply the actual date values or cash flow values inside the formula as arguments, they expect you to supply the values argument and the dates argument as arrays. (An array is just a set of numbers.)

For example, suppose that you want to calculate the internal rate of return and net present value for an investment that produces the following cash flows on the following dates:

```
1/1/2000    -1000
12/31/2000  -1000
4/15/2001   2000
12/31/2001  1000
```

If you include the actual array in the argument, you can designate the array by enclosing the values and dates arguments inside braces. To show the preceding date values in an array, for example, you would type the following:

```
{"1/1/2000","12/31/2000","4/15/2001","12/31/2001"}
```

To show the preceding cash flow values as an array, you would type the following:

```
{-1000,-1000,2000,1000}
```

If you enter the date values and cash flow values in worksheet ranges, you don't need to worry about identifying the date values and cash flow values as arrays. For example, if you enter the preceding set of date values in the worksheet range A1:A4 and the preceding set of cash flow values in the worksheet range B1:B4, you can use these worksheet ranges as the function arguments.

The XIRR and XNPV functions, predictably, require you to use date values that are valid. You must also use the same number of date values as you use cash flow values. If you supply an invalid argument or set of arguments, Excel returns the #NUM error value.

NOTE *Excel considers the first date value to be the starting date of the investment. Accordingly, the first date value in your array or worksheet range must be the earliest. Subsequent date values don't have to be in chronological order, however.*

XIRR

The XIRR function calculates the internal rate of return for an investment given its cash flows, the dates of those cash flows, and, optionally, an initial guess as to the internal rate of return. The function uses the following syntax:

```
XIRR(values,dates,guess)
```

For example, suppose that you want to calculate the internal rate of return for an investment that produces the following cash flows on the following dates:

```
1/1/2000    -1000
12/31/2000  -1000
4/15/2001   2000
12/31/2001  1000
```

To calculate the internal rate of return for this set of cash flows using the XIRR function and using a starting guess of 20%, you would use the following formula:

```
=XIRR({-1000,-1000,2000,1000},{"1/1/2000","12/31/2000","4/15/2001","12/
31/2001"},.2)
```

The formula returns the value .470251, which is equivalent to 47.0251% annually.

If the date values were stored in the worksheet range A1:A4 and the cash flow values were stored in the worksheet range B1:B4, you could instead use the following formula:

```
=XIRR(B1:B4,A1:A4,.2)
```

XNPV

The XNPV function calculates the net present value for an investment given its cash flows, the dates of those cash flows, and the annual discount rate. The function uses the following syntax:

```
XNPV(rate,values,dates)
```

For example, suppose that you want to calculate the net present value for an investment that produces the following cash flows on the following dates:

```
1/1/2000    -1000

12/31/2000  -1000

4/15/2001   2000

12/31/2001  1000
```

If the date values were stored in the worksheet range A1:A4, the cash flow values were stored in the worksheet range B1:B4, and you wanted to use a discount rate of 15%, you would use the following formula:

```
=XNPV(.15,B1:B4,A1:A4)
```

The formula returns the value 557.17.

Curiously, the XNPV function doesn't accept date values supplied as text strings. For example, although the following formula is equivalent to the preceding one, it returns the #VALUE error value:

```
=XNPV(.15,{-1000,-1000,2000,1000},{"1/1/2000","12/31/2000","4/15/
2001","12/31/2001"})
```

You could, however, rewrite this formula using equivalent serial date values (the serial date 36526 for 1/1/2000, the serial date value 36891 for 12/31/2000, and so on), and then Excel returns the correct net present value:

```
=XNPV(0.15,{-1000,-1000,2000,1000},{36526,36891,36996,37256})
```

Using the Coupon Dates Add-In Functions

Excel provides six functions that let you make coupon date calculations more easily: COUPDAYBS, COUPDAYS, COUPDAYSNC, COUPNCD, COUPNUM, and COUPPCD.

All six coupon date functions use four standard arguments: the settlement date, the maturity date, the frequency, and the basis.

The settlement date specifies the date the bond is settled, or purchased. The maturity date specifies the date the bond matures, or expires. You may enter these date arguments either as text strings enclosed in quotation marks or as serial date values.

The frequency argument gives the number of coupon payments made each year: you specify 1 to indicate an annual coupon, 2 to indicate a semiannual coupon, and 4 to indicate a quarterly coupon.

The basis argument specifies the number of days in the month and year assumed for the date calculations. You specify the basis as 0 for the US (or NASD) version of 30 days in a month and 360 days in a year; as 1 for the actual number of days in the month and actual number of days; 2 for the actual number of days in the month but 360 days in a year; 3 for the actual number of days in the month and 365 days in a year; and 4 for the European version of 30 days in a month and 360 days in a year.

NOTE *Excel uses only the integer portion of the arguments you supply to the add-in coupon date functions. If you enter an argument with decimal values, Excel truncates the argument to just its integer component.*

The coupon date functions return an error value in several predictable cases:

- If you use an invalid date, Excel returns #VALUE.
- If you use a frequency argument other than 1, 2, or 4, Excel returns #NUM.
- If you use a day-count-basis switch other than 0, 1, 2, 3, or 4, Excel returns #NUM.
- If the settlement day follows the maturity date, Excel returns #NUM.

COUPDAYBS

The COUPDAYBS function calculates the number of days from the last coupon payment date to the settlement date given the settlement date, maturity date, coupon frequency, and basis. It uses the following syntax:

```
COUPDAYBS(settlement,maturity,frequency,basis)
```

For example, suppose you want to calculate the number of days from the last coupon payment date to the settlement date in the following situation: Someone purchases a 10-year bond on November 26, 2000, with a maturity date of April 30, 2008. The bond pays coupons twice a year based on the US, or NASD, assumption. To make this calculation, you use the following formula:

```
COUPDAYBS("11/26/2000","4/30/2008",2,0)
```

The function returns the value 26.

COUPDAYS

The COUPDAYS function calculates the number of days in the coupon period that includes the settlement date given the settlement date, the maturity date, the coupon frequency, and the day count basis. It uses the following syntax:

```
COUPDAYS(settlement,maturity,frequency,basis)
```

For example, suppose you want to calculate the number of days in the coupon payment in the following situation: Someone purchases a 10-year bond on November 26, 2000, with a maturity date of April 30, 2008. The bond pays its coupon twice a year based on the US, or NASD, assumption. To make this calculation, you use the following formula:

```
=COUPDAYS("11/26/2000","4/30/2008",2,0)
```

The function returns the value 180.

COUPDAYSNC

The COUPDAYSNC function calculates the number of days from the settlement date to the next coupon date given the settlement date, the maturity date, the frequency, and the basis. It uses the following syntax:

```
COUPDAYSNC(settlement,maturity,frequency,basis)
```

For example, suppose you want to calculate the number of days from the settlement date to the next coupon payment in the following situation: Someone purchases a 10-year bond on November 26, 2000, with a maturity date of April 30, 2008. The bond pays its coupon twice a year based on the US, or NASD, assumption. To make this calculation, you use the following formula:

```
=COUPDAYSNC("11/26/2000","4/30/2008",2,0)
```

The function returns the value 154.

COUPNCD

The COUPNCD function calculates the coupon date that follows the settlement date given the settlement date, the maturity date, the frequency, and the day-count-basis switch. It uses the following syntax:

```
COUPNCD(settlement,maturity,frequency,basis)
```

For example, suppose you want to calculate the next coupon payment after the settlement date in the following situation: Someone purchases a 10-year bond on November 26, 2000, with a maturity date of April 30, 2008. The bond pays its coupon twice a year based on the US, or NASD, assumption. To make this calculation, you use the following formula:

```
=COUPNCD("11/26/2000","4/30/2008",2,0)
```

The function returns the value 37011, which is the serial date value for April 30, 2001.

NOTE *Excel uses serial values to represent dates: 1 for January 1, 1900; 2 for January 2, 1900; and so on. To format a serial date value to look like a date, select the cell, choose the Format menu's Cell command, click the Number tab, and choose a date format.*

COUPNUM

The COUPNUM function calculates the number of number of coupons, or interest payments, made between the settlement date and maturity date. The function, which rounds this result up to the nearest integer value, uses the following syntax:

```
COUPNUM(settlement,maturity,frequency,basis)
```

For example, suppose you want to calculate the number of coupons, or interest payments, in the following situation: Someone purchases a 10-year bond on November 26, 2000, with a maturity date of April 30, 2008. The bond pays its coupon twice a year based on the US, or NASD, assumption. To make this calculation, you use the following formula:

```
=COUPNUM("11/26/2000","4/30/2008",2,0)
```

The function returns the value 15.

COUPPCD

The COUPPCD function calculates the coupon date before the settlement date given the settlement date, the maturity date, the frequency, and the basis. It uses the following syntax:

```
COUPPCD(settlement,maturity,frequency,basis)
```

For example, suppose you want to calculate the coupon payment date preceding the settlement date in the following situation: Someone purchases a 10-year bond on November 26, 2000, with a maturity date of April 30, 2008. The bond pays its coupon twice a year based on the US, or NASD, assumption. To make this calculation, you use the following formula:

```
COUPPCD("11/26/2000","4/30/2008",2,0)
```

The function returns the value 36830, which is the serial date value for October 31, 2000.

NOTE *Excel uses serial values to represent dates: 1 for January 1, 1900; 2 for January 2, 1900; and so on. To format a serial date value to look like a date, select the cell, choose the Format menu's Cell command, click the Number tab, and choose a date format.*

Using the Cumulative Interest and Principal Add-In Functions

Excel supplies two functions specifically for calculating cumulative interest and principal payments on a loan: CUMIPMT, which calculates the cumulative interest payments on a bond or note, and CUMPRINC, which calculates the cumulate principal payments on a bond or note.

Both functions use the same set of arguments, including the interest rate, the loan term (or number of periods), the loan balance (or present value), the starting date and the ending date of the period for which you want to calculate the cumulative interest or principal payments, and a type-of-annuity switch to indicate whether the stream of payments occurs as an ordinary annuity or an annuity due.

The interest rate, loan term, loan balance, and type-of-annuity switch arguments work the same way for the CUMIPMT and CUMPRINC functions as they work for the standard financial functions.

NOTE *If you have questions about how the interest rate, loan term, loan balance, or type-of-annuity switch arguments work, refer to the earlier section "Using the Payment Functions."*

The starting date and ending date arguments, as noted earlier, simply provide the starting and ending points for the period of time for which you want to calculate the cumulative interest or principal paid.

Both the CUMIPMT and CUMPRINC functions return an error value in several predictable situations as well as one surprising situation:

- If the interest rate or the loan term rate is less than or equal to zero.
- If the starting date or ending date is nonsensical or if the starting date follows the ending date.
- If the loan present value is less than or equal to zero. (Note that this means you *don't* use the convention of specifying the loan present value as a negative number to show that it's a cash outflow.)

CUMIPMT

The CUMIPMT function calculates the cumulative interest paid on a loan between two dates you specify given the interest rate, loan term, loan present value, the starting date and ending date, and the type-of-annuity switch. It uses the following syntax:

```
CUMIPMT(rate,nper,pv,start period,end period,type)
```

Suppose, for example, that you want to calculate the cumulative interest paid on a $1,000,000, ten-year equipment loan that charges 9% interest and requires monthly payments arranged as an annuity due. Further suppose that you want to calculate the cumulative interest payments made over the first five years, or sixty months. To make this calculation, you use the following formula:

```
=CUMIPMT(.09/12,10*12,1000000,1,60,1)
```

The function returns the value –360094.

CUMPRINC

The CUMPRINC function calculates the cumulative principal paid on a loan between two dates you specify given the interest rate, loan term, loan present value, the starting date and ending date, and the type-of-annuity switch. It uses the following syntax:

```
CUMPRINC(rate,nper,pv,start period,end period,type)
```

Suppose, for example, that you want to calculate the cumulative principal paid on a $1,000,000, ten-year equipment loan that charges 9% interest and requires monthly payments arranged as an annuity due. Further suppose that you want to calculate the cumulative principal payments made over the first five years, or sixty months. To make this calculation, you use the following formula:

```
=CUMPRINC(.09/12,10*12,1000000,1,60,1)
```

The function returns the value –394303.

Using the Dollar Pricing Add-In Functions

Excel's DOLLARDE and DOLLARFR functions let you easily convert security dollar prices from decimal prices to fractional prices or from fractional prices to decimal prices.

DOLLARDE

The DOLLARDE function, for example, converts a fractional dollar price to an equivalent decimal price based on the fractional price and the fraction's denominator. It uses the following syntax:

```
DOLLARDE(fractional price,fraction)
```

For example, to convert the fractional price 25 $^{2}/_{16}$ to an equivalent decimal price, you use the following formula:

```
=DOLLARDE(25.02,16)
```

The function returns the value 25.125.

And to convert the fractional price 25 $^{1}/_{8}$ to an equivalent decimal price, you use the following formula:

```
=DOLLARDE(25.1,8)
```

This function returns the value 25.125.

DOLLARFR

The DOLLARFR functions converts a dollar decimal price into a dollar fractional price given the decimal price and the fraction's denominator. It uses the following syntax:

```
DOLLARFR(decimal price,fraction)
```

For example, to convert the price 10.125 to a fractional price in eighths, you use the following formula:

```
=DOLLARFR(10.125,8)
```

The function returns the value 10.1.

Similarly, to convert the price 10.125 to a fractional price in sixteenths, you use the following formula:

```
=DOLLARFR(10.125,16)
```

The function returns the value 10.02.

NOTE *When you work with the DOLLARDE and DOLLARFR functions, remember that Excel expects your fraction argument to be an integer. If it isn't, Excel uses just the integer portion. Also, Excel expects the fraction argument to be a positive value. If it isn't, Excel returns the #NUM! error value.*

Using the French Depreciation Add-In Functions

Excel supplies two special functions for making French depreciation calculations, AMORDEGRC and AMORLINC. Both functions use a similar sets of arguments, including the asset cost, purchase date, date at the end of the first period, salvage value, accounting period, depreciation rate, and basis. You specify the basis as 0 for the US (or NASD) version of 30 days in a month and 360 days in a year; as 1 for the actual number of days in the month and year; 2 for the actual number of days in the month but 360 days in a year; 3 for the actual number of days in the month and 365 days in a year; and 4 for the European version of 30 days in a month and 360 days in a year.

AMORDEGRC

The AMORDEGRC function uses the following syntax:

```
AMORDEGRC(cost,purchase date,first period,salvage,period,rate,basis)
```

For example, if you want to depreciate a piece of equipment purchased on July 15, 2000, for 3,000 French francs, using a salvage value of 600 French francs, a 15% depreciation rate, and the first accounting period ends on December 31, 2000, you use the following formula:

```
=AMORDEGRC(3000,"7/15/2000","12/31/2000",600,1,0.15,1)
```

The AMORDEGRC function returns the value 930.

You should know several points before using the AMORDEGRC function:

- The function uses different coefficients depending on the asset's estimated life. If the life of the asset falls between three and four years, the function uses a depreciation coefficient equal to 1.5. If the life of the asset falls between five and six years, the function uses a depreciation coefficient equal to 2. If the life of the asset is greater than six years, the function uses a depreciation coefficient equal to 2.5.
- To fully depreciate an asset, the depreciation rate grows to 50% for the next-to-last period and 100% for the last period.
- If you specify the estimated economic life of an asset as a non-integer value between 0 and 5—such as 4.5 for example—the AMORDEGRC function returns the #NUM! error value.

AMORLINC

The AMORLINC function uses the following syntax:

```
AMORLINC(cost,purchase date,first period,salvage,period,rate,basis)
```

For example, if you want to depreciate a piece of equipment purchased on July 15, 2000, for 3,000 French francs, using a salvage value of 600 French francs, a 15% depreciation rate, and the first accounting period ends on December 31, 2000, you use the following formula:

```
=AMORLINC(3000,"7/15/2000","12/31/2000",600,1,0.15,1)
```

The function returns the value 450.

Using the Future Value Add-In Functions

Two of the add-in functions don't really fit into one of the other categories, but because they both calculate the future value of some investment, I've grouped them together as future value functions: FVSCHEDULE and RECEIVED.

FVSCHEDULE

The FVSCHEDULE function calculates the future value of an investment given the present value of the investment and a schedule of interest rates. The function uses the following syntax:

```
FVSCHEDULE(principal,rate schedule)
```

As an example of how this function works, suppose you want to calculate the future value of an initial investment equal to $25,000 invested over the next five years at the following annual interest rates: .06, .07, .07, .08, .05. The following formula makes this calculation:

```
=FVSCHEDULE(25000,{.06,.07,.07,.08,.05})
```

If the annual interest rates are stored in the worksheet range B1:B5, as shown in Figure 5-7, you might also use the following formula:

```
=FVSCHEDULE(25000,B1:B5)
```

Both functions return the same value, 34405.39.

	A	B	C
1	Year 1 rate	0.06	
2	Year 2 rate	0.07	
3	Year 3 rate	0.07	
4	Year 4 rate	0.08	
5	Year 5 rate	0.05	
6			
7	Future Value	34405.39	
8			
9			

Figure 5-7 A worksheet set up to use the FVSCHEDULE function.

NOTE *The FVSCHEDULE function returns the #VALUE error value if you supply a nonnumeric interest rate argument. Note, however, that you can use zero or reference empty cells to show no interest.*

RECEIVED

The RECEIVED function calculates the future value amount of a fully invested, or zero-coupon, security given its settlement date, maturity date, the initial investment, the discount rate, and the basis. The function uses the following syntax:

```
RECEIVED(settlement,maturity,investment,discount,basis)
```

The settlement date specifies the date the bond is settled, or purchased. The maturity date specifies the date the bond matures, or expires. You may enter these date arguments either as text strings enclosed in quotation marks or as serial date values.

The investment is the initial amount invested, or the present value.

The discount rate specifies the annual discount rate used to price the bill.

Finally, the familiar basis argument specifies the number of days in the month and in the year assumed for the date calculations. You specify the basis as 0 for the US (or NASD) version of 30 days in a month and 360 days in a year; as 1 for the actual number of days in the month and year; 2 for the actual number of days in the month but 360 days in a year; 3 for the actual number of days in the month and 365 days in a year; and 4 for the European version of 30 days in a month and 360 days in a year.

NOTE *Excel uses only the integer portion of the arguments you supply to the add-in RECEIVED function. If you enter an argument with decimal values, Excel truncates the argument to just its integer component.*

NOTE *The RECEIVED function returns an error value if a date argument or the set of date arguments is invalid or if the investment or discount rate is set to zero.*

For example, suppose you want to calculate the future value amount received for a bond you purchase on May 1, 2000, and that matures on October 31, 2002. Further suppose that you purchased the bond for $50,000 based on a 6% discount rate. If you want to use the US (or NASD) day-count-basis assumption, you use the following formula:

```
=RECEIVED("5/1/2000","10/31/2002",50000,.06,0)
```

The function returns the value 58823.53.

Using the Interest Rate Add-In Functions

Excel provides four functions that let you make interest rate calculations: DISC, EFFECT, INTRATE, and NOMINAL.

The DISC and INTRATE functions, which are related, work from the same basic set of arguments: the settlement date, the maturity date, the redemption value, the price, the frequency, and the basis.

The settlement date specifies the date the bond is settled, or purchased. The maturity date specifies the date the bond matures, or expires. You may enter the date arguments either as text strings enclosed in quotation marks (for example, “7/4/99”) or as serial date values (for example, 37000 for April 19, 2001.)

The redemption argument is the bond’s redemption value per each $100 of face value.

The price argument shows the price of a bond expressed as a percentage of its face value. For example, a bond that cost $991.83 would be priced at 99.183.

The frequency argument gives the number of coupon payments made each year: you specify 1 to indicate an annual coupon, 2 to indicate a semiannual coupon, and 4 to indicate a quarterly coupon.

The basis argument specifies the number of days in the month and in the year assumed for the date calculations. You specify the basis as 0 for the US (or NASD) version of 30 days in a month and 360 days in a year; as 1 for the actual number of days in the month and year; 2 for the actual number of days in the month but 360 days in a year; 3 for the actual number of days in the month and 365 days in a year; and 4 for the European version of 30 days in a month and 360 days in a year.

The EFFECTIVE and NOMINAL functions, which are also related, work from a set of three arguments: the effective annual interest rate, the nominal interest rate, and the number of compounding periods in the year.

DISC

The DISC function calculates the discount rate for a security—the amount by which the redemption value is reduced expressed as an annual percentage—given its settlement date, maturity date, price, redemption, and basis. The function uses the following syntax:

```
DISC(settlement,maturity,price,redemption,basis)
```

For example, suppose you want to calculate the discount rate on a zero-coupon, $100 redemption-value bond that you purchased on July 10, 2000, for 99.875. If you choose to use the US (or NASD) day-count-basis assumption, you use the following formula to make this calculation:

```
=DISC("7/10/2000","11/30/2000",97.875,100,0)
```

The function returns the value .054643, which is equivalent to 5.4643%.

NOTE *The DISC function returns an error value if a date argument or the set of date arguments is invalid or if a bond price or redemption value is set to zero.*

EFFECT

The EFFECT function calculates the effective annual interest given the stated annual interest rate and the number of annual compounding periods. The function uses the following syntax:

```
EFFECT(nominal rate,compounding periods)
```

For example, if you want to calculate the effective interest rate when the nominal rate is 6%, but this rate is compounded daily (based on a 360-day year), you use the following formula:

```
=EFFECT(.06,360)
```

The function returns the value .061831, which is equivalent to 6.1831%.

NOTE *The EFFECT function returns an error value if you supply nonnumeric arguments, a nominal rate argument equal to 0, or a number of compounding periods argument equal to some value less than 1.*

INTRATE

The INTRATE function calculates the interest rate for a fully invested, or zero-coupon, security given its settlement date, maturity date, the initial investment amount, the redemption value, and the basis. The function uses the following syntax:

```
INTRATE(settlement,maturity,investment,redemption,basis)
```

For example, suppose you want to calculate the interest rate on a zero-coupon, $100 redemption-value bond that you purchased on July 10, 2000, for 99.875. If you choose to use the US (or NASD) day-count basis assumption, you use the following formula to make this calculation:

```
=INTRATE("7/10/2000","11/30/2000",97.875,100,0)
```

The function returns the value .055829, which is equivalent to 5.5829%.

NOTE *The INTRATE function returns an error value if a date argument or the set of date arguments is invalid or if the investment or redemption value is set to zero.*

NOMINAL

The function calculates the nominal annual interest given the effective annual interest rate and the number of annual compounding periods. The function uses the following syntax:

```
NOMINAL(effective rate,compounding periods)
```

For example, if you want to calculate the nominal interest rate when the effective rate is 6.1831% and this rate is based on daily compounding (based on a 360-day year), you use the following formula:

```
=NOMINAL(.061831,360)
```

The function returns the value .06, which is equivalent to 6%.

NOTE *The EFFECT function returns an error value if you supply nonnumeric arguments, a nominal rate argument equal to 0, or a number of compounding periods argument equal to some value less than 1.*

Using the Price and Yield Add-In Functions

Excel provides 10 functions that let you make discount, yield, and price calculations for securities such as bonds more easily: ODDFPRICE, ODDFYIELD, ODDLPRICE, ODDLYIELD, PRICE, PRICEDISC, PRICEMAT, YIELD, YIELDDISC, and YIELDMAT.

NOTE *Excel's online help file supplies the actual formulas used for many of these yield and price functions.*

As a group, the 10 yield and price functions use a set of standard arguments: the settlement date, the maturity date, the frequency, and the basis.

The yield and price functions use several standard date arguments, for example. The settlement date specifies the date the bond is settled, or purchased. The maturity date specifies the date the bond matures, or expires. The issue date is the date on which a security is issued. You may enter the date arguments either as text strings enclosed in quotation marks (for example, "7/4/99") or as serial date values (for example, 37000 for April 19, 2001.)

The functions for pricing odd-period securities—ODDFPRICE, ODDFYIELD, ODDLPRICE, and ODDLYIELD—also require the date of the first regular coupon payment or the date of the last regular coupon payment in order to calculate the first or last odd period.

The rate argument is the bond's interest rate. The yield argument is the bond's annual yield.

The redemption argument is the bond's redemption value per each $100 of face value.

The price argument shows the price of a bond expressed as a percentage of its face value. For example, a bond that cost $991.83 would be priced at 99.183.

The frequency argument gives the number of coupon payments made each year: you specify 1 to indicate an annual coupon, 2 to indicate a semiannual coupon, and 4 to indicate a quarterly coupon.

The basis argument specifies the number of days in the month and year assumed for the date calculations. You specify the basis as 0 for the US (or NASD) version of 30 days in a month and 360 days in a year; as 1 for the actual number of days in the month and year; 2 for the actual number of days in the month but 360 days in a year; 3 for the actual number of days in the month and 365 days in a year; and 4 for the European version of 30 days in a month and 360 days in a year.

NOTE *Excel uses only the integer portion of the arguments you supply to the add-in price and yield date functions. If you enter an argument with decimal values, Excel truncates the argument to just its integer component.*

The yield and price functions return the #NUM error value in several predictable cases:

- If you use an invalid date, Excel returns #VALUE. Note that this means your date arguments must make sense collectively, too. For example, your maturity date must follow the settlement date.
- If you use a frequency argument other than 1, 2, or 4, Excel returns #NUM.
- If you use a day-count-basis switch other than 0, 1, 2, 3, or 4, Excel returns #NUM.
- If the rate or yield or price is less than zero, Excel returns #NUM.

ODDFPRICE

The ODDFPRICE function calculates the price per $100 face value of a security when the first period is odd—shorter or longer than a typical coupon period—given the settlement date, maturity date, issue date, first coupon date, coupon rate, yield, redemption price, coupon frequency, and basis. It uses the following syntax:

```
ODDFPRICE(settlement,maturity,issue,first
coupon,rate,yield,redemption,frequency,basis)
```

Suppose, for example, that you want to calculate the price of an odd-period bond that you purchased on March 4, 2000, that will mature on May 31, 2011, was originally issued on December 7, 1999, pays a semiannual coupon of 3.5% starting on November 30, 2000, and is priced to yield 6.5% annually. Further assume that you want to use the European, 30-days-in-a-month, 360-days-in-a-year day count basis. To make this calculation, you use the following formula:

```
=ODDFPRICE("3/4/2000","5/31/2011","12/7/1999","11/30  2000",
      .035*2,.065,100,2,4)
```

The function returns 100.5063.

ODDFYIELD

The ODDFYIELD function calculates the yield of a security when the first period is odd—shorter or longer than a typical coupon period—given the settlement date, maturity date, issue date, first coupon date, coupon rate, price, redemption price, coupon frequency, and basis. It uses the following syntax:

```
ODDFYIELD(settlement,maturity,issue,first  coupon,rate,price,redemption,
frequency,basis)
```

Suppose, for example, that you want to calculate the price on an odd-period bond that you purchased on March 4, 2000, that will mature on May 31, 2011, was originally issued on December 7, 1999, pays a semiannual coupon of 3.5% starting on November 30, 2000, and is priced at 99.183 but will be redeemed at face value. Further assume that you want to use the European, 30-days-in-a-month, 360-days-in-a-year day count basis. To make this calculation, you use the following formula:

```
=ODDFYIELD("3/4/2000","5/31/2011","12/7/1999","11/30/2000",
      .035*2,99.183,100,2,4)
```

The function returns 0.066599, which is equivalent to 6.6599%.

ODDLPRICE

The ODDLPRICE function calculates the price per $100 face value of a security when the last period is odd—shorter or longer than a typical coupon period—given the settlement date, maturity date, issue date, last coupon date, coupon rate, yield, redemption price, coupon frequency, and basis. It uses the following syntax:

```
ODDLPRICE(settlement,maturity,last
coupon,rate,yield,redemption,frequency,basis)
```

Suppose, for example, that you want to calculate the price on an odd-period bond that you purchased on March 4, 2000, that will mature on May 31, 2011, was originally issued on December 7, 1998, pays a semiannual coupon of 3.5%, last paid interest on November 30, 1999, and is priced to yield 6.5% annually. Further assume that you want to use the European, 30-days-in-a-month, 360-days-in-a-year day count basis. To make this calculation, you use the following formula:

```
=ODDLPRICE("3/4/2000","5/31/2011","11/30/1999",.035*2,.065,100,2,4)
```

The function returns 102.4757.

ODDLYIELD

The ODDFYIELD function calculates the yield of a security when the last period is odd—shorter or longer than a typical coupon period—given the settlement date, maturity date, issue date, last coupon date, coupon rate, price, redemption price, coupon frequency, and basis. It uses the following syntax:

```
ODDLYIELD(settlement,maturity,issue,last  coupon,rate,price,redemption,
frequency,basis)
```

Suppose, for example, that you want to calculate the price on an odd-period bond that you purchased on March 4, 2000, that will mature on May 31, 2011, was originally issued on December 7, 1999, pays a semiannual coupon of 3.5%, last paid a coupon on November 30, 1999, and is priced at 99.183 but will be redeemed at face value. Further assume that you want to use the European, 30-days-in-a-month, 360-days-in-a-year day count basis. To make this calculation, you use the following formula:

```
=ODDLYIELD("3/4/2000","5/31/2011","11/30/1999",.035*2,99.183,100,2,4)
```

The function returns 0.070019, which is equivalent to 7.0019%.

PRICE

The PRICE function calculates the price per $100 face value of a security given the settlement date, maturity date, coupon rate, yield, redemption price, coupon frequency, and basis. It uses the following syntax:

```
PRICE(settlement,maturity,rate,yield,redemption,frequency,basis)
```

Suppose, for example, that you want to calculate the price on a bond that you purchased on March 4, 2000, that will mature on May 31, 2011, pays a semiannual coupon of 3.5%, is priced to yield 6.5% annually, and will be redeemed at face value, or 100. Further assume that you want to use the European, 30-days-in-a-month, 360-days-in-a-year day count basis. To make this calculation, you use the following formula:

```
=PRICE("3/4/2000","5/31/2011",.035*2,.065,100,2,4)
```

The function returns 103.9299.

PRICEDISC

The PRICEDISC function calculates the price per $100 face value of a discounted security given the settlement date, maturity date, discount rate, redemption price, and basis. It uses the following syntax:

```
PRICEDISC(settlement,maturity,discount,redemption,basis)
```

Suppose, for example, that you want to calculate the price on a discounted security that you purchased on March 4, 2000, that will mature on May 31, 2011, is discounted using a rate of 6.5% annually, and will be redeemed at face value, or 100. Further assume that you want to use the European, 30-days-in-a-month, 360-days-in-a-year day count basis. To make this calculation, you use the following formula:

```
=PRICEDISC("3/4/2000","5/31/2011",.065,100,4)
```

The function returns 26.9472.

PRICEMAT

The PRICEMAT function calculates the price per $100 face value of a security that will pay its interest upon maturity given the settlement date, maturity date, issue date, coupon rate, yield, and basis. It uses the following syntax:

```
PRICEMAT(settlement,maturity,issue,rate,yield,basis)
```

Suppose, for example, that you want to calculate the price on a security that you purchased on March 4, 2000, was first issued on March 4, 1999, that will mature on May 31, 2011, pays a coupon of 3.5% semiannually, and is discounted using a rate of 6.5% annually. Further assume that you want to use the European, 30-days-in-a-month, 360-days-in-a-year day count basis. To make this calculation, you use the following formula:

```
=PRICEMAT("3/4/2000","5/31/2011","3/4/1999",.035*2,.065,4)
```

The function returns 100.2923.

YIELD

The YIELD function calculates the yield of a security given the settlement date, maturity date, coupon rate, price, redemption price, coupon frequency, and basis. It uses the following syntax:

```
YIELD(settlement,maturity,rate,price,redemption,frequency,basis)
```

Suppose, for example, that you want to calculate the yield on a bond that you purchased on March 4, 2000, that will mature on May 31, 2011, pays a semiannual coupon of 3.5%, is priced at 101.1425, and will be redeemed at face value, or 100. Further assume that you want to use the European, 30-days-in-a-month, 360-days-in-a-year day count basis. To make this calculation, you use the following formula:

```
=YIELD("3/4/2000","5/31/2011",.035*2,101.1425,100,2,4)
```

The function returns 0.068507, which is equivalent to 6.8507%.

YIELDDISC

The YIELDDISC function calculates the yield of a discounted security given the settlement date, maturity date, price, redemption price, and basis. It uses the following syntax:

```
YIELDDISC(settlement,maturity,price,redemption,basis)
```

Suppose, for example, that you want to calculate the yield on a discounted security that you purchased on March 4, 2000, that will mature on May 31, 2011, is discounted at 56.1762, and will be redeemed at face value, or 100. Further assume that you want to use the European, 30-days-in-a-month, 360-days-in-a-year day count basis. To make this calculation, you use the following formula:

```
=YIELDDISC("3/4/2000","5/31/2011",56.1762,100,4)
```

The function returns 0.069412, which is equivalent to 6.9412%.

YIELDMAT

The YIELDMAT function calculates the yield of a security that will pay its interest upon maturity given the settlement date, maturity date, issue date, coupon rate, price, and basis. It uses the following syntax:

```
YIELDMAT(settlement,maturity,issue,rate,price,basis)
```

Suppose, for example, that you want to calculate the yield on a security that you purchased on March 4, 2000, was first issued on March 4, 1999, that will mature on May 31, 2011, pays a coupon of 3.5% semiannually, and is priced at 95.8194. Further assume that you want to use the European, 30-days-in-a-month, 360-days-in-a-year day count basis. To make this calculation, you use the following formula:

```
=YIELDMAT("3/4/2000","5/31/2011","3/4/1999",.035*2,95.8194,4)
```

The function returns 0.071698, which is equivalent to 7.1698%.

Using the Treasury Bill Add-In Functions

Excel provides three add-in financial functions for analyzing United States Treasury bills: TBILLEQ, which calculates the bond-equivalent yields; TBILLPRICE, which calculates the price of a Treasury bill; and TBILLYIELD, which calculates the yield on a Treasury bill.

The Treasury bill functions use a set of standard arguments: the settlement date, the maturity date, the discount rate, and the price. The settlement date specifies the date the bill is settled, or purchased. The maturity date specifies the date the bill matures, or expires. (You may enter these date arguments either as text strings enclosed in quotation marks or as serial date values.) The discount rate specifies the annual discount rate used to price the bill. The price specifies the price per $100 of face value.

NOTE *Excel uses only the integer portion of the arguments you supply to the Treasury bill functions. If you enter an argument with decimal values, Excel truncates the argument to just its integer component.*

All three functions return an error value if the settlement or maturity date isn't a valid date, if the discount rate is less than zero, if the settlement date falls after the maturity date, or if the maturity date isn't within one year of the settlement date.

TBILLEQ

The TBILLEQ function calculates the bond-equivalent yield for a Treasury bill given its settlement date, maturity date, and a discount rate. It uses the following syntax:

```
TBILLEQ(settlement,maturity,discount)
```

For example, if you want to calculate the equivalent bond yield on a Treasury bill if the settlement date is April 8, 2001, the maturity date is July 15, 2001, and the discount rate is 3%, you use the following formula:

```
=TBILLEQ("4/8/2001","7/15/2001",.03)
```

The function returns the value .03067, or 3.067%.

TBILLPRICE

The TBILLPRICE function calculates the price per $100 of face value for a Treasury bill given the settlement date, the maturity date, and the discount rate. It uses the following syntax:

```
TBILLPRICE(settlement,maturity,discount rate)
```

For example, if you want to calculate the price on a Treasury bill if the settlement date is April 8, 2001, the maturity date is July 15, 2001, and the discount rate is 3%, you use the following formula:

```
=TBILLPRICE("4/8/2001","7/15/2001",.03)
```

The function returns the value 99.1833, which means that you would pay $99.1833 for each $100 of Treasury bill face value.

TBILLYIELD

The TBILLYIELD function calculates the yield delivered by a Treasury bill given the settlement date, maturity date, and price. It uses the following syntax:

```
TBILLYIELD(settlement,maturity,price)
```

For example, if you want to calculate the yield on a Treasury bill if the settlement date is April 8, 2001, the maturity date is July 15, 2001, and the price is 99.1833, you use the following formula:

```
=TBILLYIELD("4/8/2001","7/15/2001",99.1833)
```

The function returns the value 0.0302482, which is equivalent to 3.0248%.